

-- SegMap.Mesa Edited by Sandman on April 12, 1978 3:34 PM

DIRECTORY

IODefs: FROM "iodefs",
 AltoDefs: FROM "altodefs",
 SegmentDefs: FROM "segmentdefs";

DEFINITIONS FROM AltoDefs, IODefs, SegmentDefs;

SegMap: PROGRAM IMPORTS IODefs, SegmentDefs SHARES SegmentDefs = PUBLIC BEGIN

byte: NumberFormat = NumberFormat[8,FALSE,TRUE,3];
 word: NumberFormat = NumberFormat[8,FALSE,TRUE,6];

PrintDataSegment: PROCEDURE [seg:DataSegmentHandle] RETURNS [BOOLEAN] =
 BEGIN OPEN seg;
 WriteNumber[VMpage,byte]; WriteChar[SP];
 WriteNumber[AddressFromPage[VMpage],word];
 WriteString[" "L];
 WriteNumber[pages,byte];
 WriteLine[" VM"L];
 WriteChar[CR];
 RETURN[~More[]]
 END;

PrintFileSegment: PROCEDURE[seg:FileSegmentHandle] RETURNS [BOOLEAN] =
 BEGIN OPEN seg;
 WriteNumber[VMpage,byte]; WriteChar[SP];
 WriteNumber[AddressFromPage[VMpage],word]; WriteChar[SP];
 WriteNumber[base,byte]; WriteChar[SP];
 WriteNumber[pages,byte]; WriteChar[SP];
 WriteString["SN"L]; WriteOctal[file.fp.serial.part2];
 SELECT class FROM
 code => WriteString[" code"L];
 ENDCASE;
 WITH seg SELECT FROM
 remote => WriteString[" remote"L];
 ENDCASE;
 IF read OR write THEN WriteChar[' '];
 IF read THEN WriteChar['R'];
 IF write THEN WriteChar['W'];
 IF swappedin THEN WriteString[" in"L];
 IF lock > 0 THEN
 BEGIN
 WriteString[" lock="L];
 WriteOctal[lock];
 END;
 WriteChar[CR];
 RETURN[~More[]]
 END;

lc: INTEGER;
 full: INTEGER = 18;

More: PROCEDURE RETURNS [BOOLEAN] =
 BEGIN c: CHARACTER;
 IF (lc ← lc+1) >= full THEN
 BEGIN lc ← 0;
 DO -- until non-random input
 SELECT (c ← ReadChar[]) FROM
 SP,CR,LF => EXIT;
 DEL => RETURN[FALSE];
 ENDCASE;
 ENDLOOP;
 END;
 RETURN[TRUE]
 END;

DO WriteChar[CR]; lc ← 0;
 [] ← EnumerateFileSegments[PrintFileSegment];
 WriteChar[CR]; lc ← full;
 IF More[] THEN
 [] ← EnumerateDataSegments[PrintDataSegment];
 WriteChar[CR]; STOP;
 ENDLOOP;

END.